Minds are like parachutes. They only function when they are open. – Sir James Dewar (1842-1923)

ABSTRACT
Macro programming is generally considered an advanced topic. But, while macros certainly can be challenging, it is also true that the basic concepts are not difficult to learn. This paper is designed for people who know the basics of SAS programming, but know nothing about SAS macro programming. We explain how the macro processor works, and how to use macros and macro variables. Using these techniques you can create flexible, reusable code that can save you time and effort.

WHY USE MACROS?
Because macro code takes longer to write and debug than standard SAS code, you generally won’t use macros in programs that will be run only a few times. But if you find yourself writing similar code over and over again, then macros may make your job easier.

Macros can help in several ways. First, with macros you can make one small change in your program and have SAS echo that change throughout your program. Second, macros can allow you to write a piece of code and use it over and over again in the same program or in different programs. Third, you can make your programs data driven, letting SAS decide what to do based on actual data values.

THE MACRO PROCESSOR
The most important concept to keep in mind whenever you write macro code is that

You are writing a program that writes a program.

Here is how it works.

```
macro statements
macro processor
```

```
standard SAS statements
```

When you submit a standard SAS program, SAS compiles and then immediately executes it. But when you write macro code, there is an extra step. Before SAS can compile and execute your program, SAS must pass your macro statements to the macro processor which then “resolves” your macros generating standard SAS code. Because you are writing a program that writes a program, this is sometimes called meta-programming.

MACROS VS. MACRO VARIABLES
As you construct macro programs, you will use two basic building blocks: macros and macro variables. You can tell them apart because the names of macro variables start with an ampersand (&), while the names of macros start with a percent sign (%). A macro variable is like a standard data variable except that it does not belong to a data set and has only a single value which is always character. The value of a macro variable could be a variable name, a numeral, or any text you want substituted in your program.

A macro, however, is a larger piece of a program that can contain complex logic including complete DATA and PROC steps, and macro statements such as %IF-%THEN-%ELSE and %DO-%END. Macros often—but not always—contain macro variables.

THINK GLOBALLY AND LOCALLY
Macro variables come in two varieties: either local or global. A macro variable’s scope is local if it is defined inside a macro. Its scope is global if it is defined in “open code” which is everything outside a macro. You can use a global macro variable anywhere in your program, but you can use a local macro variable only inside its own macro.

If you keep this in mind as you write your programs, you will avoid two common mistakes: trying to use a local macro variable outside its own macro, and accidentally creating local and global macro variables having the same name.

YOU MAY QUOTE ME ON THAT
Another caveat to keep in mind is that the macro processor doesn’t check for macros inside single quotes. To get around this, simply use double quotes for titles or other quoted strings that contain macro code.

SUBSTITUTING TEXT WITH %LET
%LET is a straightforward macro statement that simply assigns a value to a macro variable. Despite its simplicity, by learning this one statement you will greatly increase the flexibility of your programs.

Suppose you have a program that you run once a month. Every time you have to edit the program so it will select...
data for the correct month and print the correct title. This is
time-consuming and prone to errors. You can use %LET
to create a macro variable. Then you only have to change
the value of the macro variable in the %LET statement,
and SAS will repeat that new value throughout your
program.

The general form a %LET statement is

```plaintext
%LET macro-variable-name = value;
```

where `macro-variable-name` is a name you make up
following the standard rules for SAS names (32 characters
or fewer in length; starting with a letter or underscore; and
containing only letters, numerals, or underscores). `Value`
is the text to be substituted for the macro variable name, and
can be up to 64,000 characters long. The following
statements each create a macro variable.

```plaintext
%LET iterations = 5;
%LET winner = Lance Armstrong;
```

Notice that there are no quotation marks around `value`
even when it contains characters. Blanks at the beginning
and end will be trimmed, and everything else between the
equal sign and semicolon will become part of the value for
the macro variable.

To use the macro variable, you simply add the ampersand
prefix (`&`) and stick the macro variable name wherever you
want its value to be substituted.

```plaintext
DO i = 1 to &iterations;
TITLE "First: &winner";
```

After being resolved by the macro processor, these
statements would become

```plaintext
DO i = 1 to 5;
TITLE "First: Lance Armstrong";
```

Example

A company that manufactures bicycles maintains a file
listing all their models. For each model they record its
name, class (Road, Track, or Mountain), list price, and
frame material. Here is a subset of the data:

<table>
<thead>
<tr>
<th>Model</th>
<th>Class</th>
<th>Price</th>
<th>Frame</th>
</tr>
</thead>
<tbody>
<tr>
<td>Black Bora</td>
<td>Track</td>
<td>796</td>
<td>Aluminum</td>
</tr>
<tr>
<td>Delta Breeze</td>
<td>Road</td>
<td>399</td>
<td>CroMoly</td>
</tr>
<tr>
<td>Jet Stream</td>
<td>Track</td>
<td>1130</td>
<td>CroMoly</td>
</tr>
<tr>
<td>Mistral</td>
<td>Road</td>
<td>1995</td>
<td>Carbon Comp</td>
</tr>
<tr>
<td>Nor'easter</td>
<td>Mountain</td>
<td>899</td>
<td>Aluminum</td>
</tr>
<tr>
<td>Santa Ana</td>
<td>Mountain</td>
<td>459</td>
<td>Aluminum</td>
</tr>
<tr>
<td>Scirocco</td>
<td>Mountain</td>
<td>2256</td>
<td>Titanium</td>
</tr>
<tr>
<td>Trade Wind</td>
<td>Road</td>
<td>759</td>
<td>Aluminum</td>
</tr>
</tbody>
</table>

This DATA step reads the raw data from a file named
Models.dat creating a SAS data set named MODELS.

```plaintext
DATA models;
INFILE 'c:\MyRawData\Models.dat';
TRUNCOVER;
INPUT Model $ 1-12 Class $ Price $ 28-38;
RUN;
```

Frequently the sales people want a list showing all the
models of one particular class. Sometimes they want a list
of road bikes, but other times they want mountain or track
bikes. You could write three different programs—one for
each type—or you could write one program with a macro
variable. This program uses a %LET statement to create a
macro variable named &BIKECLASS. To run the program,
you type in the class of bike you want at the beginning of
the program, and then SAS echoes that value throughout
the program.

```plaintext
%LET bikeclass = Mountain;
* Use a macro variable to subset;
PROC PRINT DATA = models NOOBS;
   WHERE Class = "&bikeclass";
   FORMAT Price DOLLAR6.;
   TITLE "Current Models of " &bikeclass " Bicycles";
RUN;
```

Note that the macro variable &BIKECLASS will be global
because it is created outside of a macro in open code.
When you submit the program with a value of “Mountain”,
then the macro processor will resolve the macro variable
and create this standard SAS code.

```plaintext
* Use a macro variable to subset;
PROC PRINT DATA = models NOOBS;
   WHERE Type = "Mountain";
   FORMAT Price DOLLAR6.;
   TITLE "Current Models of " "Mountain Bicycles";
RUN;
```

Here are the results:

<table>
<thead>
<tr>
<th>Current Models of Mountain Bicycles</th>
</tr>
</thead>
<tbody>
<tr>
<td>Model</td>
</tr>
<tr>
<td>----------</td>
</tr>
<tr>
<td>Nor'easter</td>
</tr>
<tr>
<td>Santa Ana</td>
</tr>
<tr>
<td>Scirocco</td>
</tr>
</tbody>
</table>

This was a short program, with only two occurrences of the
macro variable. But imagine if you had a long program
with dozens of occurrences of the macro variable sprinkled
throughout. You could save a lot of time and trouble by
changing the macro variable only once at the beginning.

**Creating Modular Code with Macros**

Anytime you find yourself repeating the same program
statements over and over, you might want to consider
creating a macro instead. Macros are simply a group of
SAS statements that have a name. And, anytime you
want that group of statements in your program, you use
the name instead of the re-typing all the statements.

The general form of a macro is

```plaintext
%MACRO macro-name;
macro-text
%MEND;
```

Example
The %MACRO statement tells SAS that this is the beginning of the macro and the %MEND statement signals the end of the macro. **Macro-name** is a name you make up for your macro. The name must follow standard SAS naming conventions (start with a letter or underscore; contain only letters, numerals or underscores; and can be up to 32 characters in length). You don’t need to specify the **macro-name** in the %MEND statement, but it will make your programs easier to understand if you include it. **Macro-text** represents the SAS statements that you want in your macro.

After you define your macro, you need to invoke it when you want to use it. Do this by adding a percent sign in front of the macro-name like this:

```sas
%macro-name
```

While you do not need to end this statement with a semicolon, it generally does no harm either.

**Example**

Looking again at the bicycle models data, the sales personnel like to have lists of models sorted both by model name and by price. The following program creates a macro named PRINTIT that has the PROC PRINT statements you need for the report. Then the program calls the macro twice; first without sorting the data (it is already sorted by model name), and then after executing a PROC SORT by Price.

```sas
%MACRO printit;
PROC PRINT DATA = models NOOBS;
  TITLE 'Current Models';
  VAR Model Class Frame Price;
  FORMAT Price DOLLAR6.;
RUN;
%MEND printit;

%printit

PROC SORT DATA = models;
  BY Price;
%printit
```

Here are the standard SAS statements that are generated by the macro processor. The first PROC PRINT is generated by the first call to the PRINTIT macro, the PROC SORT comes from the original program and the second PROC PRINT is generated by the second call to the PRINTIT macro.

```sas
PROC PRINT DATA = models NOOBS;
  TITLE 'Current Models';
  VAR Model Class Frame Price;
  FORMAT Price DOLLAR6.;
RUN;

PROC SORT DATA = models;
  BY Price;
PROC PRINT DATA = models NOOBS;
  TITLE 'Current Models';
  VAR Model Class Frame Price;
  FORMAT Price DOLLAR6.;
RUN;
```

And here are the results of the two PROC PRINTs: one before sorting and one after sorting by Price.

<table>
<thead>
<tr>
<th>Model</th>
<th>Class</th>
<th>Frame</th>
<th>Price</th>
</tr>
</thead>
<tbody>
<tr>
<td>Black Bora</td>
<td>Track</td>
<td>Aluminum</td>
<td>$796</td>
</tr>
<tr>
<td>Delta Breeze</td>
<td>Road</td>
<td>CroMoly</td>
<td>$399</td>
</tr>
<tr>
<td>Jet Stream</td>
<td>Track</td>
<td>CroMoly</td>
<td>$1,130</td>
</tr>
<tr>
<td>Mistral</td>
<td>Road</td>
<td>Carbon Comp</td>
<td>$1,995</td>
</tr>
<tr>
<td>Nor’easter</td>
<td>Mountain</td>
<td>Aluminum</td>
<td>$899</td>
</tr>
<tr>
<td>Santa Ana</td>
<td>Mountain</td>
<td>Aluminum</td>
<td>$459</td>
</tr>
<tr>
<td>Scirocco</td>
<td>Mountain</td>
<td>Titanium</td>
<td>$2,256</td>
</tr>
<tr>
<td>Trade Wind</td>
<td>Road</td>
<td>Aluminum</td>
<td>$759</td>
</tr>
</tbody>
</table>

This program is fairly simple with only two calls to the macro, so you didn’t save much by putting the PROC PRINT statements in a macro. But if you had many calls to the macro, you could simplify your program a great deal by using a macro. And, if you need to make a change to any of the PROC PRINT statements, if they are in a macro, then you would only need to make one change instead of trying to find every occurrence of the PROC PRINT and risk making mistakes.

### ADDING PARAMETERS TO MACROS

Macros allow you to execute a set of SAS statements with just one statement, and while this alone can be helpful, macros are even more powerful when you add parameters to them. Parameters are macro variables whose values are set when you invoke the macro. Parameters give your macros flexibility.

To add parameters to your macro, simply list the macro-variable names followed by an equal sign in parentheses after the macro name:

```sas
%MACRO macro-name (parameter-1=, parameter-2=, . . . parameter-n=);
macro-text
%MEND macro-name;
```

For example, if you have a macro named %MONTHLYREPORT that has parameters for the month and region, it might start like this:

```sas
%MONTHLYREPORT (month=, region=);
```

```sas
PROC PRINT DATA = models NOOBS;
  TITLE 'Current Models';
  VAR Model Class Frame Price;
  FORMAT Price DOLLAR6.;
RUN;
```
%MACRO monthlyreport (month=, region=);

Then when you invoke the macro, give the values for the macro variables after the equal sign:

%monthlyreport (month=May, region=West);

Because these macro variables are defined in the macro, they are local to the macro and you cannot use them in any SAS statements outside the macro.

Example

The PRINTIT macro from the previous example prints the desired reports, but you have to sort the data between calls to the macro and the title does not reflect the sort order of the report. This new macro, SORTANDPRINT includes the PROC SORT statements and adds parameters so that you can vary both the sort variable and the sort sequence (descending or not). Then the macro variables are added to a title statement, so the report states how it is sorted. There are two calls to the macro, producing two reports.

* For sortvar enter the variable for sorting the report. For sortseq use either Descending or do not enter a value for an ascending sequence;

%MACRO sortandprint(sortseq=, sortvar=);
PROC SORT DATA = models;
   BY &sortseq &sortvar;
PROC PRINT DATA = models NOOBS;
   TITLE 'Current Models';
   TITLE2 "Sorted by &sortseq &sortvar";
   VAR Model Class Frame Price;
   FORMAT Price DOLLAR6.;
RUN;
%MEND sortandprint;

%sortandprint(sortseq=Descending, sortvar=Price);
%sortandprint(sortseq=, sortvar=Class);

The first call to the macro gives the &SORTSEQ macro variable the value DESCENDING and the &SORTVAR macro variable the value Price. So the statements generated by this call will sort the data by descending values of the variable Price. The second call to the macro gives &SORTSEQ no value and &SORTVAR the value Class. So this call will produce statements that sort the data by ascending (the default) Class. Here are the SAS statements that are generated by the macro processor:

PROC SORT DATA = models;
   BY DESCENDING Price;
PROC PRINT DATA = models NOOBS;
   TITLE 'Current Models';
   TITLE2 "Sorted by Descending Price";
   VAR Model Class Frame Price;
   FORMAT Price DOLLAR6.;
RUN;

Here are the results of the two PROC PRINTs, one sorted by descending Price, the other by Class.

<table>
<thead>
<tr>
<th>Model</th>
<th>Class</th>
<th>Frame</th>
<th>Price</th>
</tr>
</thead>
<tbody>
<tr>
<td>Scirocco</td>
<td>Mountain</td>
<td>Titanium</td>
<td>$2,256</td>
</tr>
<tr>
<td>Mistral</td>
<td>Road</td>
<td>Carbon Comp</td>
<td>$1,995</td>
</tr>
<tr>
<td>Jet Stream</td>
<td>Track</td>
<td>CroMoly</td>
<td>$1,130</td>
</tr>
<tr>
<td>Nor'easter</td>
<td>Mountain</td>
<td>Aluminum</td>
<td>$899</td>
</tr>
<tr>
<td>Black Bora</td>
<td>Track</td>
<td>Aluminum</td>
<td>$796</td>
</tr>
<tr>
<td>Trade Wind</td>
<td>Road</td>
<td>Aluminum</td>
<td>$759</td>
</tr>
<tr>
<td>Santa Ana</td>
<td>Mountain</td>
<td>Aluminum</td>
<td>$459</td>
</tr>
<tr>
<td>Delta Breeze</td>
<td>Road</td>
<td>CroMoly</td>
<td>$399</td>
</tr>
</tbody>
</table>

MPRINT system option

We have been showing you what SAS sees after the macro processor has resolved your program, but normally you won't see these statements. However if you use the MPRINT system option in your program, then SAS will print the resolved statements in the SAS log. This can be very useful for debugging purposes. To turn on the MPRINT option, submit an OPTIONS statement like this:

OPTIONS MPRINT;
Here is what the SAS log looks like with the MPRINT option turned on:

```sas
%MACRO sortandprint(sortseq=, sortvar=);
   PROC SORT DATA=models;
     BY &sortseq &sortvar;
   PROC PRINT DATA=models NOOBS;
     TITLE 'Current Models';
     TITLE2 "Sorted by &sortseq &sortvar";
     VAR Model Class Frame Price;
     FORMAT Price DOLLAR6.;
   RUN;
%MEND sortandprint;

%sortandprint(sortseq=Descending, sortvar=Price);

MPRINT(SORTANDPRINT):   PROC SORT DATA=models;
MPRINT(SORTANDPRINT):   BY Descending Price;
NOTE: There were 8 observations read from the data set WORK.MODELS.
NOTE: The data set WORK.MODELS has 8 observations and 4 variables.
NOTE: PROCEDURE SORT used (Total process time):
     real time           0.55 seconds
     cpu time            0.05 seconds

MPRINT(SORTANDPRINT):   PROC PRINT DATA=models NOOBS;
MPRINT(SORTANDPRINT):   TITLE 'Current Models';
MPRINT(SORTANDPRINT):   TITLE2 "Sorted by Descending Price";
MPRINT(SORTANDPRINT):   VAR Model Class Frame Price;
MPRINT(SORTANDPRINT):   FORMAT Price DOLLAR6.;
MPRINT(SORTANDPRINT):   RUN;
NOTE: There were 8 observations read from the data set WORK.MODELS.
NOTE: PROCEDURE PRINT used (Total process time):
     real time           0.81 seconds
     cpu time            0.07 seconds

%sortandprint(sortseq=, sortvar=Class);

MPRINT(SORTANDPRINT):   PROC SORT DATA=models;
MPRINT(SORTANDPRINT):   BY Class;
NOTE: There were 8 observations read from the data set WORK.MODELS.
NOTE: The data set WORK.MODELS has 8 observations and 4 variables.
NOTE: PROCEDURE SORT used (Total process time):
     real time           0.04 seconds
     cpu time            0.03 seconds

MPRINT(SORTANDPRINT):   PROC PRINT DATA=models NOOBS;
MPRINT(SORTANDPRINT):   TITLE 'Current Models';
MPRINT(SORTANDPRINT):   TITLE2 "Sorted by Class";
MPRINT(SORTANDPRINT):   VAR Model Class Frame Price;
MPRINT(SORTANDPRINT):   FORMAT Price DOLLAR6.;
MPRINT(SORTANDPRINT):   RUN;
NOTE: There were 8 observations read from the data set WORK.MODELS.
NOTE: PROCEDURE PRINT used (Total process time):
     real time           0.09 seconds
     cpu time            0.00 seconds

RUN;
```
You can see that SAS has inserted into the regular SAS log the MPRINT lines. The statements generated by the macro processor are all labeled with the word MPRINT followed by the name of the macro that generated the statements—in this case SORTANDPRINT. By using the MPRINT system option it is easy to see the standard SAS statements your macro is generating.

**CONDITIONAL LOGIC**

With macros and macro variables, you have a lot of flexibility. You can increase that flexibility still more by using conditional macro statements such as %IF. Here are the general forms of statements used for conditional logic in macros:

```sas
%IF condition %THEN action;
%ELSE %IF condition %THEN action;
%ELSE action;
%IF condition %THEN %DO;
    action;
%END;
```

These statements probably look familiar because there are parallel statements in standard SAS code, but don’t confuse these with their standard counterparts. These statements can only be used inside a macro, and can perform actions that would be completely impossible for standard SAS statements. With %IF, actions can include other macro statements or even complete DATA and PROC steps. Remember, the macro statements won’t appear in the standard SAS code generated by the macro processor because you are writing a program that writes a program.

**Automatic macro variables**

Every time you invoke SAS, the macro processor automatically creates certain macro variables. You can use these in your programs. Two of the most commonly used automatic variables are:

<table>
<thead>
<tr>
<th>Variable Name</th>
<th>Example</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>&amp;SYSDATE</td>
<td>31OCT03</td>
<td>Character value of the date that job or session began</td>
</tr>
<tr>
<td>&amp;SYSDAY</td>
<td>Friday</td>
<td>Day of the week that job or session began</td>
</tr>
</tbody>
</table>

**Example**

The company maintains a file with information about every order they receive. For each order, the data are the customer ID number, date the order was placed, model name, and quantity ordered. Here are the data:

<table>
<thead>
<tr>
<th>Customer ID</th>
<th>Order Date</th>
<th>Model</th>
<th>Quantity</th>
</tr>
</thead>
<tbody>
<tr>
<td>287</td>
<td>15OCT03</td>
<td>Delta Breeze</td>
<td>15</td>
</tr>
<tr>
<td>287</td>
<td>15OCT03</td>
<td>Santa Ana</td>
<td>15</td>
</tr>
<tr>
<td>274</td>
<td>16OCT03</td>
<td>Jet Stream</td>
<td>1</td>
</tr>
<tr>
<td>174</td>
<td>17OCT03</td>
<td>Santa Ana</td>
<td>20</td>
</tr>
<tr>
<td>174</td>
<td>17OCT03</td>
<td>Nor'easter</td>
<td>5</td>
</tr>
<tr>
<td>174</td>
<td>17OCT03</td>
<td>Scirocco</td>
<td>1</td>
</tr>
<tr>
<td>347</td>
<td>18OCT03</td>
<td>Mistral</td>
<td>1</td>
</tr>
<tr>
<td>287</td>
<td>21OCT03</td>
<td>Delta Breeze</td>
<td>30</td>
</tr>
<tr>
<td>287</td>
<td>21OCT03</td>
<td>Santa Ana</td>
<td>25</td>
</tr>
</tbody>
</table>

This DATA step reads the raw data from a file named Orders.dat creating a SAS data set named ORDERS.

```sas
DATA orders;
  INFILE 'c:\MyRawData\Orders.dat';
  INPUT CustomerID $ 1-3 OrderDate DATE7. Model $ 13-24 Quantity;
RUN;
```

Every Monday the president of the company wants a detail-level report showing all the current orders. On Friday the president wants a report summarized by customer. You could write two separate programs to run these reports, or you could write one program using macros.

Combining conditional logic with the automatic variable &SYSDAY, the following program creates a macro named REPORTS that prints a detail report if you run it on Monday, and a summary report if you run it on Friday (and no report on other days).

```sas
%MACRO reports;
  %IF &SYSDAY = Monday %THEN %DO;
    PROC PRINT DATA = orders NOOBS;
    FORMAT OrderDate DATE7.;
    TITLE "&SYSDAY Report: " "Current Orders";
  %END;
  %ELSE %IF &SYSDAY = Friday %THEN %DO;
    PROC TABULATE DATA = orders;
    CLASS CustomerID;
    VAR Quantity;
    TABLE CustomerID ALL, Quantity;
    TITLE "&SYSDAY Report: Summary " "of Orders";
  %END;
%MEND;
```

If you run this program on Monday, the macro processor with generate the following standard SAS code.

```sas
PROC PRINT DATA = orders NOOBS;
  FORMAT OrderDate DATE7.;
  TITLE "Monday Report: " "Current Orders";
RUN;
```

And the output will look like this:

<table>
<thead>
<tr>
<th>Customer ID</th>
<th>Order Date</th>
<th>Model</th>
<th>Quantity</th>
</tr>
</thead>
<tbody>
<tr>
<td>287</td>
<td>15OCT03</td>
<td>Delta Breeze</td>
<td>15</td>
</tr>
<tr>
<td>287</td>
<td>15OCT03</td>
<td>Santa Ana</td>
<td>15</td>
</tr>
<tr>
<td>274</td>
<td>16OCT03</td>
<td>Jet Stream</td>
<td>1</td>
</tr>
<tr>
<td>174</td>
<td>17OCT03</td>
<td>Santa Ana</td>
<td>20</td>
</tr>
<tr>
<td>174</td>
<td>17OCT03</td>
<td>Nor'easter</td>
<td>5</td>
</tr>
<tr>
<td>174</td>
<td>17OCT03</td>
<td>Scirocco</td>
<td>1</td>
</tr>
<tr>
<td>347</td>
<td>18OCT03</td>
<td>Mistral</td>
<td>1</td>
</tr>
<tr>
<td>287</td>
<td>21OCT03</td>
<td>Delta Breeze</td>
<td>30</td>
</tr>
<tr>
<td>287</td>
<td>21OCT03</td>
<td>Santa Ana</td>
<td>25</td>
</tr>
</tbody>
</table>
If you run this program on Friday, the macro processor will generate the following standard SAS code.

```sas
PROC TABULATE DATA = orders;
   CLASS CustomerID;
   VAR Quantity;
   TABLE CustomerID ALL, Quantity;
   TITLE "Friday Report: Summary of Orders";
```

And the output will look like this:

<table>
<thead>
<tr>
<th>CustomerID</th>
<th>Sum</th>
</tr>
</thead>
<tbody>
<tr>
<td>174</td>
<td>26.00</td>
</tr>
<tr>
<td>274</td>
<td>1.00</td>
</tr>
<tr>
<td>287</td>
<td>85.00</td>
</tr>
<tr>
<td>347</td>
<td>1.00</td>
</tr>
<tr>
<td>All</td>
<td>113.00</td>
</tr>
</tbody>
</table>

**Friday Report: Summary of Orders**

---

**DATA-DRIVEN MACROS**

This is where your macro programs begin to take on a life of their own. Using CALL SYMPUT you can let a macro program look at the data and then decide for itself what to do.

To assign a value to a single macro variable, use CALL SYMPUT with this general form:

```sas
CALL SYMPUT("macro-variable", value);
```

where *macro-variable* is the name of a macro variable, either new or old, and is enclosed in quotes. *Value* is the value you want to assign to that macro variable. *Value* can be the name of a variable or it can be a constant enclosed in quotes.

CALL SYMPUT is often used in IF-THEN statements, for example

```sas
IF Place = 1 THEN
   CALL SYMPUT("Status", "Winner");
ELSE CALL SYMPUT("Status", "Other");
```

These statements create a macro variable named &STATUS and assign to it a value of either Winner or Other based on the value of the variable Place. The following CALL SYMPUT uses a variable name as its value.

```sas
IF Place = 1 THEN
   CALL SYMPUT("WinningTime", Time);
```

This statement tells SAS to create a macro variable named &WINNINGTIME and set it equal to the current value of the variable TIME when the value of Place is 1.

You cannot create a macro variable with CALL SYMPUT and use it in the same DATA step. Here's why. When you submit macro code, it is resolved by the macro processor, and then compiled and executed. Not until the final stage—execution—does SAS see your data. CALL SYMPUT takes a data value from the execution phase, and passes it back to the macro processor for use in a later step. That's why you must put CALL SYMPUT in one DATA step but not use it until a later step.

**Example**

The marketing department wants a report showing all the orders placed by their best customer. The following program finds the customer with the single biggest order and then prints all the orders for that customer.

```sas
*Sort by Quantity;
PROC SORT DATA = orders;
   BY DESCENDING Quantity;
*Use CALL SYMPUT to find biggest order;
PROC TABULATE DATA = orders;
   CLASS CustomerID;
   VAR Quantity;
   TABLE CustomerID ALL, Quantity;
TITLE "Customer &biggest Had the Single Largest Order";
RUN;
```

This program has three steps. First PROC SORT sorts the data by descending Quantity so that the first observation is the largest single order.

Then a DATA step reads the sorted data, and in the first iteration of the DATA step (when _N_ = 1), uses CALL SYMPUT to assign the current value of CustomerID (287) to the macro variable &BIGGEST. The keyword _NULL_ in the DATA statement tells SAS not to bother creating a new SAS data set, and the STOP statement tells SAS to stop after the first iteration. The DATA _NULL_ and STOP are not necessary but they make the program more efficient by preventing SAS from reading and writing observations that won't be used.

The final step takes the macro variable &BIGGEST and inserts its value into the PROC PRINT so that SAS selects just the orders for customer number 287. Here are the standard SAS statements that SAS will see after the macro processor has resolved the final step:
"Print all obs for customer with biggest order;
PROC PRINT DATA = orders NOOBS;
   WHERE CustomerID = "287";
   FORMAT OrderDate DATE7.;
   TITLE "Customer 287 Had the "
   "Single Largest Order";
RUN;

The output will look like this:

<table>
<thead>
<tr>
<th>Customer ID</th>
<th>Order Date</th>
<th>Model</th>
<th>Quantity</th>
</tr>
</thead>
<tbody>
<tr>
<td>287</td>
<td>21OCT03</td>
<td>Delta Breeze</td>
<td>30</td>
</tr>
<tr>
<td>287</td>
<td>21OCT03</td>
<td>Santa Ana</td>
<td>25</td>
</tr>
<tr>
<td>287</td>
<td>15OCT03</td>
<td>Delta Breeze</td>
<td>15</td>
</tr>
<tr>
<td>287</td>
<td>15OCT03</td>
<td>Santa Ana</td>
<td>15</td>
</tr>
</tbody>
</table>

CONCLUSIONS

There is no doubt that macros can be very complicated, but don’t let that scare you off. Learning to use macro variables and incorporating simple macros into your programs can also make your work a lot easier. Get your feet wet by starting with adding macro variables to your programs. Then when you get the hang of macro variables, start building simple macros.

To avoid mangling your macros, always write them one piece at a time. First write your program in standard SAS code. When that is working and bug-free, then add your %MACRO and %MEND statements. When that is working, then add your parameters, if any, one at a time.

REFERENCES


ABOUT THE AUTHORS

Lora Delwiche and Susan Slaughter are also the authors of The Little SAS Book: A Primer published by SAS Institute, and may be contacted at:

Susan J. Slaughter (530) 756-8434
susan@avocetsolutions.com

Lora D. Delwiche (530) 795-3772
ldelwiche@onramp113.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

1 There are a few exceptions. %INCLUDE, for example, is not part of the macro facility despite its percent sign prefix. And it is possible, though rather confusing, to write macros that do not start with a percent sign.